MODELING AND MIGRATION WITH THE MONOCHROMATIC WAVE EQUATION -- VARIABLE VELOCITY AND ATTENUATION

Einar Kjartansson

Abstract

A review of the theory for the 45-degree monochromatic wave equation leads to a simple scheme for migration and diffractions that can readily handle lateral variations in velocity. Anelasticity can be included without a change in the finite difference algorithm. Sample Fortran programs are given for both modeling and migration of zero-offset sections for arbitrary velocity and Q structures, with synthetic examples. The algorithm has been adapted for large datasets by taking advantage of the speed of the SEP array processor.

Introduction

While the theory for the finite difference modeling of waves in the frequency domain has been developed in detail (Claerbout, 1970, 1976), its use appears to have been rather limited. As the properties of the earth may be considered time invariant for the duration of seismic experiments and linear at seismic amplitudes, no generality is lost by Fourier transforming over time. There are several advantages in working in the frequency domain. Each Fourier component of the seismogram may be propagated separately, which can simplify manipulations of large datasets compared to time-domain methods. A time shift over a non-integer number of sampling intervals consists in the frequency domain of a simple multiplication. Perhaps the greatest advantage of the frequency domain is that all time-derivatives are evaluated exactly by a simple multiplication. This becomes increasingly important as more accurate equations involving higher time-derivatives are used. This also makes it possible to include the effects of anelasticity, often at little or no additional cost.

Theory

It is shown by Claerbout (1976, p. 196) that the scalar wave equation

$$P_{zz} + P_{xx} = \frac{1}{v^2} P_{tt} \tag{1}$$

becomes

$$Q_{zz} + Q_{xx} + 2\, i\, \bar{m}\, Q_z + (m^2 - \bar{m}^2)Q = 0 \tag{2}$$

when Q is defined by

$$Q(x,z,\omega) = e^{i\bar{m}z}\, \bar{P}(x,z,\omega) \tag{3}$$

$$\bar{P}(x,z,\omega) = \int_{-\infty}^{\infty} P(x,z,t)\, e^{-i\omega t}\, dt \tag{4}$$

$$m = -\frac{\omega}{v} \tag{5}$$

$$\bar{m} = -\frac{\omega}{\bar{v}} \tag{6}$$

In the derivation of this result, $\bar{v}$ has been assumed to be independent of x and z, while v may be a function of both x and z. Since we are using a minus sign in the forward Fourier transform (Bracewell, 1965), the signs in Equations (5) and (6) are different from those used by Claerbout (1976).

The Q_{zz} term is eliminated if each term in Equation (2) is differentiated with respect to z, multiplied by $i/2\bar{m}$, and added to the original equation. The result is

$$\frac{i}{2\bar{m}} Q_{zzz} + \frac{i}{2\bar{m}} Q_{xxz} + Q_{xx} + 2\, i\, \bar{m}\, Q_z + \frac{i}{2\bar{m}} (m^2 - \bar{m}^2)Q_z + (m^2 - \bar{m}^2)Q + \frac{im}{\bar{m}} \frac{dm}{dz} Q = 0 \tag{7}$$

It should be noted that we have not made any approximations yet: Equation (7) is simply the scalar wave equation in a shifted coordinate frame.

For waves traveling in approximately the same direction as the coordinate frame is shifted, Q_{zzz} should be small in relation to the other terms. If it is dropped, the result

$$\frac{i}{2\bar{m}} Q_{xxz} + Q_{xx} + 2\, i\, \bar{m}\, Q_z + \frac{i}{2\bar{m}} (m^2 - \bar{m}^2) Q_z +$$

$$(m^2 - \bar{m}^2) Q + \frac{im}{\bar{m}} \frac{dm}{dz} Q = 0 \qquad (8)$$

will be first order in z. The dispersion relation is shown in Figure 1. As shown in Figure 1, the dispersion relation for Equation (8) starts to deviate significantly from the correct value when $\bar{m}$ and m differ by more than 50 percent.

In many cases this is not a serious limitation. I have used Equation (8) extensively to model seismic sections, without any difficulties. If it is assumed that the coefficients in Equation (1) are locally constant, one may set $\bar{m} = m$ and then apply a time shift at each depth step. This assumption that the coefficients are locally constant is commonly made, e.g. when a wave equation is derived to satisfy a particular dispersion relation.

Thus one can get the wavefield $\bar{P}(z + \Delta z)$ from $\bar{P}(z)$ by setting

$$Q(z) = \bar{P}(z) \qquad (9)$$

and solving

$$\frac{i}{2m(x,z)} Q_{xxz} + Q_{xx} + 2\, i\, m(x,z)\, Q_z = 0 \qquad (10)$$

for $Q(z + \Delta z)$, and then time shifting:

$$\bar{P}(z + \Delta z) = e^{im(x,z)\Delta z}\ Q(z + \Delta z)$$

Equation (10) is simpler to code than Equation (8) and has a more accurate dispersion relation, but does not treat the effects of velocity gradients as accurately as Equation (8). Both equations fit on the finite difference star described by Claerbout (1976, pp. 184-189) and may be solved using the Crank-Nicolson scheme. A different perspective on Equation (10) is offered by Brown (this report, p. 214).

Implementation

The modeling and migration of zero-offset seismic sections is commonly approximated by the exploding reflector model; that is, waves originate at the reflectors at $t = 0$ and propagate toward the surface with one-half the true velocity of the medium. Thus, when modeling a zero-offset section on the computer one can start with a blank upgoing wavefield below the lowest reflector and use Equations (9), (10) and (11) to continue the wavefield up toward the surface. Since a delta function at $t = 0$ has a Fourier transform that is simply a constant, independent of frequency, one can then model the exploding reflectors by adding the reflection coefficient to all of the frequencies at each z-step. The time section is then obtained by inversely Fourier transforming the results at the surface.

Migration of zero-offset data is simply the inverse of the above. One starts by Fourier transforming the time section and then continuing each frequency down, using either a negative v or Δz. The value of the wavefield at $t = 0$ is then extracted at each depth by summing over the real part of all the frequencies. An optional step that removes the effect of the wraparound in the FFT is to subtract the value of the reflector from the wavefield.

Both modeling and migration can be done by taking one frequency at a time through all the z-steps, or taking all the frequencies through one z-step at a time. It is only possible to subtract the reflectors from the migrated wavefield, however, when all the frequencies are taken together. In situations where both the reflector and velocity map, and the Fourier transform of the wavefield, are too large to fit in the main memory of the computer, disk IO is minimized by using some combination of the above -- that is, either taking as many frequencies as will fit into memory through all the z-steps or keeping as much of the velocity and reflector structure as fits in memory while taking all the frequencies through that part of the structure. Using this last arrangement we have been able to take full advantage of the speed of the SEP array processor (Newkirk and Claerbout, SEP-14, p. 285; Thorson, this report, p. 275) for migrations of several hundred traces of COCORP data (see Bloxsom and Ottolini, this report, p. 251). In fact, the simplicity of the data handling of monochromatic equations is so important for the particular hardware configuration of the SEP computer that the turnaround for variable velocity finite difference migrations is comparable to that of migrations using

the constant velocity Stolt algorithm (Ottolini, SEP-14, p. 281).

Attenuation

A first-order property of all materials, especially rocks, is the absorption of elastic energy and the resulting change in the shape of transient waveforms. Most available data is consistent with the assumption that the energy is absorbed by a linear process, and that the energy loss per cycle is independent of frequency. Elsewhere I have shown (Kjartansson, 1978) that these conditions are satisfied by a model that implies a complex, frequency-dependent velocity of the form

$$v = v_0(i\,\omega)^{\gamma} \tag{12}$$

where γ is related to the seismic quality factor Q by

$$\frac{1}{Q} = \tan(\pi\,\gamma) \tag{13}$$

The possible range for γ is $0 < \gamma < \frac{1}{2}$ and for Q is $\infty > Q > 0$. The limiting cases correspond to classical elasticity and Newtonian viscosity. Since the coefficients in the Crank-Nicolson scheme are complex, even for a purely elastic model, the only additional computation that results from the substitution of Equation (12) into either (5) or (6) is in the computation of the coefficients.

Appendix A is a Fortran listing of an in-core version of a zero-offset diffraction program that can handle arbitrary velocity and Q structures, and Appendix B is a listing of the corresponding migration program. Except for the input and output routines, these programs should run on other Fortran systems. Figures 2 through 5 show examples of outputs produced by these programs as well as the large dataset array processor versions.

Discussion

Figure 6 shows the impulse response for wave propagation through a material with a frequency-independent Q, for four different values of Q. It is routine practice (e.g. Burdick and Helmberger, 1978) in the computation

of synthetic earthquake seismograms to compute a seismogram for a purely elastic earth model, and to convolve the result with a response function of the kind shown in Figure 6. This is valid when all the arrivals present in the seismogram have suffered the same amount of attenuation, but is not even approximately valid for reflection seismograms unless it is assumed that all the attenuation takes place in the near-surface layers. Since the waveforms, especially at shorter periods, are often dominated by the attenuation impulse response, it seems worthwhile to include attenuation in the modeling of seismic sections.

Similarly, removal of the attenuation effects, along with the diffractions, in the migration of seismic data should help isolate the path independent source waveform and thus contribute to increasing the resolution of the results. The removal of attenuation effects is an inherently unstable process, especially in the presence of noise, so careful filtering of the high frequencies is required, and the results are likely to be sensitive to the quality and processing history of the data.

Although an understanding of the seismic attenuation may help us get sharper pictures of the subsurface, that is not the only reason for trying to measure and model it. There are both laboratory (Winkler and Nur, 1978) and theoretical reasons (Kjartansson and Denlinger, 1977; Mavko and Nur, 1978) to believe that there is some unique information about the lithology and such parameters as the temperature, porosity, porepressure and the amount of saturation, that can be extracted from a knowledge of the seismic attenuation parameters, especially when integrated with other geophysical information. None of the methods that have been discussed in this paper is applicable to the problem of estimating Q directly from data. A number of methods for estimation of Q from data have been discussed in the literature, including spectral ratios as well as time-domain methods advocated by Gladwin and Stacey (1974) and Kjartansson (1978). The ability to compute accurate synthetic seismograms for trial models of the Q structure should be valuable in comparing the various methods for estimating Q and establishing the validity of the results.



FIGURE 1.--The dispersion relation for Equation (8). Each plot shows the k_z/m as a function of k_x/m. For comparison the semi-circle is also shown. The accuracy of Equation (8) is quite acceptable for $0.7m < \bar{m} < 1.5m$.



2a



2b



2c



2d

FIGURE 2.--Modelling and migration using the programs listed in the appendices. The zero offset section observed at the surface from the reflector shown in 2d, for the velocity structure shown in 2c, is shown in 2a. Anelasticity with $Q = 20$ was used. The velocity at a unit frequency is 1 in the upper layer and 2 below. The result of a migration of the section in 2a is shown in 2b. Most of the loss in resolution is because anelasticity was included in the forward calculation, but not in the migration. Parameters used were as follow: 128 timepoints, 64 traces, 64 depthpoints, Δt of 0.06, Δz of 0.06 and Δx of 0.1.



FIGURE 3.--A zero offset section for the reflector and velocity structure shown, with Q = 40. Total time displayed is four seconds. Velocities shown are for a reference frequency of approximately 1 Hz. The calculation included 330 traces, 320 depth steps and 512 complex frequencies or 1024 timepoints. A gain proportional to $\sqrt{t}$ was applied before plotting. The traces are clipped to 1/5 of the maximum. This model was taken from Western Geophysical's Depth Migration brochure.



FIGURE 4.--Same results as in Figure 3, but in order to more realistically simulate the bandwidth of real data the traces have been differentiated with respect to time. The broadening of the waveform, caused by anelasticity is clearly shown. A gain proportional to $t^{3/2}$ was applied before plotting. The traces are clipped at 1/5 of maximum.



FIGURE 5.--The result of migrating the timesection shown in Figure 3. Total depth displayed is 4.4 km. The vertical smearing is mostly due to the fact that anelasticity was not included in the migration. Some of the lateral smearing is caused by the dipfiltering (see Appendix A), and near the ends, absorption at the side boundaries. The plot is clipped at 1/3 of maximum.



FIGURE 6.--The impulse response for a plane wave in a constant Q medium. Each plot shows the seismogram resulting from a unit impulse plane-wave source at a distance $x = \lambda C_0$, where C_0 is the phase velocity at reference frequency of 1 radian/second.

APPENDIX A

A source listing of DIFF, a 45-degree finite difference modelling program

This program with subroutines listed in Appendix C was used to generate the section shown in Figure 2a.

```
c
c       Finite difference modeling program, that
c       uses the monochromatic wave equation:
c
c
c
c          i
c        ---- Q     +     Q     +     2 m i  Q        =   0
c        2 m    xxz         xx                 z
c
c
c       Velocity, anelasticity and reflectivity may be
c       arbitrary functions of x and z.
c       An improved approximation for the second derivative
c       is used (variable beta ), see FGDP p. 222.
c       A zero offset time section is obtained by inversely
c       Fourier transforming the output of this program.
c       Dip filtering is included (see FGDP p. 225 )
c
c          Einar Kjartansson, September 1978.
c
        complex wave(64,64),t(64),d(64),a(64),b(64),e(64),f(64)
        complex aa(64),bb(64)
        complex cv0(64), cexp,cmplx
        complex m,shift,cc3,cc1,rr3,rr1,bab,ra,dipflt
        real q1(64),vel(64),ref(64),gam(64)
c
c       Read in parameters and set constants
c
        call rdparm(nom,nx,nz,dom,dx,dz,vis)
        rr1 = (0.,.5)/dz
        rr3 = (0.,2.)*dx*dx/dz
        dipflt = (0.,1.)*vis
        beta = .14
c
c       Clear upgoing wavefield
c
        do 20 iom = 1,nom
           do 20 ix = 1,nx
20            wave (ix,iom) = (0.,0.)
c
c       Take the wavefield up through the structure
c
        do 100 izinv = 1,nz
```

```
            iz = nz - izinv + 1
c
c           Get velocity, 1/Q and reflectivity
c
            call rdvst(iz,nx,vel)
            call rdqst(iz,nx,q1)
            call rdrst(iz,nx,ref)
            do 40 ix = 1,nx
               gampi =  atan(q1(ix))
               cv0(ix) = cexp((0.,-.5)*gampi)/vel(ix)
40             gam(ix) = gampi/3.141592654
            do 100 iom = 2,nom
               om = (iom-1)*dom
c
c              Apply time shift and
c              compute coefficients
c
               do 50 ix = 1,nx
                  m = -om**(1.-gam(ix))*cv0(ix)
                  shift = cexp((0.,1.)*m*dz)
                  t(ix) = shift*(wave(ix,iom)+ref(ix))
                  m = m + dipflt/vel(ix)
                  cc3 = rr3*m
                  cc1 = rr1/m + beta*cc3
                  aa(ix)= (.5,0.) - cc1
                  a(ix) = aa(ix)- (1.,0.)
                  bb(ix)  = cc1 + cc1 - (1.,0.) - cc3
50                b(ix) = bb(ix) + (2.,0.)
c
c              Absorbing side condition
c
               bab = m*cmplx(0.,dx*.25)
               ra = ((1.,0.)+bab) / ((1.,0.)-bab)
               b(1) = b(1) + ra*a(1)
               bb(1) = bb(1) + ra*aa(1)
               b(nx) = b(nx) + ra*a(nx)
               bb(nx) = bb(nx) + ra*aa(nx)
c
c              Solve Crank-Nicolson matrix equation
c
               d(1) = bb(1)*t(1) + aa(1)*t(2)
               d(nx) = bb(nx)*t(nx) + aa(nx)*t(nx-1)
               do 70 ix = 2, nx-1
70                d(ix) = bb(ix)*t(ix) + aa(ix)*(t(ix-1)+t(ix+1))
               call cvtri(a,b,a,nx,t,d,e,f)
               do 100 ix = 1,nx
100               wave(ix,iom) = t(ix)
c
c        Output the result
c
         call wrwave(nx,nom,wave)
         stop
         end
```

APPENDIX B

A source listing of DDMIG, a 45-degree finite difference migration program

This program with subroutines listed in Appendix C was used to migrate the section shown in Figure 2a, to obtain the result shown in Figure 2b. The finite difference scheme is stable when there are no lateral variations in velocity. The only instabilities encountered so far have been for migrations using velocity models with large (50% or more), near vertical discontinuities.

```
c
c
c       Finite difference migration program, that
c       uses the monochromatic wave equation:
c
c
c          i
c        ---- Q      +      Q      +      2 m i  Q          =    0
c        2 m    xxz           xx                   z
c
c
c       Velocity may be an arbitrary function of x and z.
c       Anelasticity is not included in this program.
c       An improved approximation for the second derivative
c       is used (variable beta ), see FGDP p. 222.
c       The input to this program is the Fourier tranform of a
c       zero-offset section.
c       Dip filtering is included ( see FGDP p. 225 )
c
c          Einar Kjartansson, September 1978.
c
        complex wave(64,64),t(64),d(64),a(64),b(64),e(64),f(64)
        complex aa(64),bb(64),ref(64)
        complex  cexp,cmplx
        complex shift,cc3,cc1,rr3,rr1,bab,ra,dipflt
        real m,vel(64)
c
c       Read in parameters and set constants
c
        call rdparm(nom,nx,nz,dom,dx,dz,vis)
        rr1 = (0.,.5)/dz
        rr3 = (0.,2.)*dx*dx/dz
        dipflt = (0.,1.)*vis
        beta = .14
c
c       Read the Fourier transform of the surface wavefield.
c
        call rdwave(nx,nom,wave)
c
c       Continue the wavefield down
c
        do 150 iz = 1,nz
c
c          Get the velocity and clear the reflector sum
```

```
c           The velocity is taken to be negative in migration
c
            call rdvst(iz,nx,vel)
            do 40 ix = 1,nx
40             ref(ix) = (0.,0.)
            do 100 iom = 2,nom
               om = (iom-1)*dom
c
c              Apply time shift and
c              compute coefficients
c
               do 50 ix = 1,nx
                  m = om/vel(ix)
                  shift = cexp((0.,1.)*m*dz)
                  t(ix) = shift*wave(ix,iom)
                  m = m + dipflt/vel(ix)
                  cc3 = rr3*m
                  cc1 = rr1/m + beta*cc3
                  aa(ix)= (.5,0.) - cc1
                  a(ix) = aa(ix)- (1.,0.)
                  bb(ix)  = cc1 + cc1 - (1.,0.) - cc3
50                b(ix) = bb(ix) + (2.,0.)
c
c              Absorbing side condition
c
               bab = m*cmplx(0.,dx*.25)
               ra = ((1.,0.)+bab) / ((1.,0.)-bab)
               b(1) = b(1) + ra*a(1)
               bb(1) = bb(1) + ra*aa(1)
               b(nx) = b(nx) + ra*a(nx)
               bb(nx) = bb(nx) + ra*aa(nx)
c
c              Solve Crank-Nicolson matrix equation
c
               d(1) = bb(1)*t(1) + aa(1)*t(2)
               d(nx) = bb(nx)*t(nx) + aa(nx)*t(nx-1)
               do 70 ix = 2, nx-1
70                d(ix) = bb(ix)*t(ix) + aa(ix)*(t(ix-1)+t(ix+1))
               call cvtri(a,b,a,nx,t,d,e,f)
               do 100 ix = 1,nx
c
c                 Sum to get wavefield at t = 0.
c
                  ref(ix) = ref(ix) + t(ix)
100               wave(ix,iom) = t(ix)
            do 110 ix = 1,nx
c
c              Subtract wavefield at t = 0 to remove wraparound
c
               ref(ix) = ref(ix)/nom
               do 110 iom = 1,nom
110               wave(ix,iom) = wave(ix,iom) - ref(ix)
150         call wrref(iz,nx,ref)
         stop
         end
```

APPENDIX C

A source listing of the subroutines used with DDIFF and DDMIG generate the results shown in Figure 2.

In order to run those programs on other computer systems, it should only be necessary to modify the input and output routines.

```
      subroutine rdparm(nom,nx,nz,dom,dx,dz,vis)
c                Suboutine to generate parameters.
      nom = 64
      nx = 64
      nz = 64
      dt = .06
      dom = 2.*3.141592654/(nom*dt)
      dx = .1
      dz = .06
      vis = dom
      return
      end
      subroutine rdvst(iz,nx,vel)
c                Subroutine to generate velocity model
      real vel(nx)
      do 10 ix = 1,nx
10    vel(ix) = 1.
      do 20 ix = 1,iz
20    vel(ix) = 2.
      return
      end
      subroutine rdqst(iz,nx,q1)
c                Subroutine to generate Q model
      real q1(nx)
      do 10 ix = 1,nx
10    q1(ix) = 1./20.
      return
      end
      subroutine rdrst(iz,nx,ref)
c                Subroutine to generate reflector structure.
      real ref(nx)
      if (iz .ne. 48 ) goto 20
      do 10 ix = 1,nx
      xx = (nx + 1.)*.5 - ix
10    ref(ix) = exp(-1.*xx*xx)
      return
20    do 30 ix = 1,nx
30    ref(ix) = 0.
      return
      end
      subroutine rdwave(nx,nom,wave)
c                Subroutine to read in the Fourier transformed wavefield
      complex wave(64,64)
      integer uopen, uread
      logical*1 fn(100)
```

```
      call fname('.frq',fn)
      if = uopen(fn,0)
      do 10 iom = 1,nom
10    ir = uread(if,wave(1,iom),512)
      return
      end
      subroutine wrwave(nx,nom,wave)
c             Subroutine to write on disk the wavefield.
      complex wave(64,80)
      integer uwrite,ucreat
      logical*1 fn(100)
      call fname('.frq',fn)
      if = ucreat(fn,"0664)
      do 10 iom = 1,nom
10    ir = uwrite(if,wave(1,iom),512)
      return
      end
      subroutine wrref(iz,nx,ref)
c             Subroutine to write reflector structure on disk.
      complex ref(nx)
      real rref(64)
      logical*1 fn(100)
      integer ucreat,uwrite
      if (iflag .eq. 1) goto 20
      iflag = 1
      call fname('.rst',fn)
      if = ucreat(fn,"0644)
20    do 30 ix = 1,nx
30    rref(ix) = ref(ix)
      nw = uwrite(if,rref,nx*4)
      return
      end
      subroutine cvtri(a,b,c,n,t,d,e,f)
c     Solve a tridiagonal matrix equation with
c     complex and variable coefficients
      implicit complex ( a-h,o-z)
      dimension t(n),d(n),f(n),e(n),a(n),b(n),c(n)
      n1 = n-1
      e(1) = -a(1)/b(1)
      f(1) = d(1)/b(1)
      do 10 i = 2,n1
      den = b(i)+c(i)*e(i-1)
      e(i) = -a(i)/den
10    f(i) = (d(i) - c(i)*f(i-1))/den
      t(n) = (d(n)- c(n)*f(n1))/(b(n)+c(n)*e(n1))
      do 20 j = 1,n1
      i = n-j
20    t(i) = e(i) *t(i+1) + f(i)
      return
      end
```

REFERENCES

BRACEWELL, R. (1965), *The Fourier Transform and its Applications* (New York: McGraw-Hill).

BURDICK, L. J. and D. V. HELMBERGER (1978), "The upper mantle P velocity structure of the Western United States," *J. Geophys. Res.*, 83, pp. 1699-1712.

CLAERBOUT, J. F. (1970), "Coarse grid calculations of waves in inhomogeneous media with application to delineation of complicated seismic structure," *Geophysics*, 35:3.

CLAERBOUT, J. F. (1976), *Fundamentals of Geophysical Data Processing* (New York: McGraw-Hill).

GLADWIN, M. T. and F. D. STACEY (1974), "Anelastic degradation of acoustic pulses in rock," *Phys. Earth. Planet. Int.*, 8, pp. 332-336.

KJARTANSSON, E. (1978), "Constant Q -- Wave propagation and attenuation," submitted to *J. Geophys. Res.*

KJARTANSSON, E. and R. DENLINGER (1977), "Seismic wave attenuation due to thermal relaxation in porous media," presented at the 47th Annual International SEG Meeting, September 21, 1977, in Calgary.

MAVKO, G. M. and A. NUR (1978), "Wave attenuation in partially saturated rocks," *Geophysics* in press.

WINKLER, K. and A. NUR (1978), "Attenuation in dry and water saturated Massilon sandstone," presented at the 48th Annual Internation SEG Meeting, San Francisco, November 1, 1978.